
What is Agena ?

� Agena is an interpreted procedural programming language.

� It can be used in scientific, scripting, and many other applications.

� Its syntax looks like very simplified Algol 68 with elements taken from 
Maple, Lua and SQL, and some other languages.

� Binaries are available for Solaris, Mac OS X, Windows, OS/2 – ArcaOS, 
Linux, Raspberry Pi, and DOS.

� Agena is Open Source, thus it is free.

� The implementation is based on the ANSI C sources of Lua 5.1.

� Sources and binaries are available at:

http://agena.sourceforge.net

Getting Started



Installing Agena

� In Solaris, OS/2 – ArcaOS, Linux, Windows, and Mac OS X, the 
respective installer automatically installs and sets up Agena. 
You do not have to add further settings yourself after installing the 
binaries.

� Information on how to install the DOS and Windows portable version is 
included in the manual or the respective read.me files.

Running Agena

� In Windows and OS/2 - ArcaOS, simply click the       icon in the 
programme group to start the interpreter.

� In Solaris, Linux, Mac and DOS, type agena in a shell. 

� Statements can be entered right after the '> ' prompt.

AgenaEdit, 1

� AgenaEdit is an editor providing syntax-highlighting and a runtime 
environment for Windows, Solaris and Linux. It can be started by
entering agenaedit in a shell.

AgenaEdit, 2

� Type your programme in the editor window and press F5 to run it.

� Mark consecutive lines in your programme with a mouse or the keyboard 
and press F6 to execute only these lines.

� During computation, press the `break` button to interrupt the current 
computation.

� Press the `restart` button to clear all variables.

� Save or open your programmes using the `File` menu in the editor
window.

� Just browse through the menu items for the other features. 

12



First Steps, 1

� Any valid Agena code can be entered at the console with or without a 
trailing colon or semicolon:

� If an expression or statement is finished with a colon, it will be 
evaluated and its value printed at the console. (This is not supported 
in AgenaEdit, use the print function instead.)

� If the expression ends with a semicolon or neither with a colon nor a 
semicolon, it will be evaluated, but nothing is printed.

� You may optionally insert one or more white spaces between operands 
in your statements.

� Assume you would like to add the numbers 1 and 2 and show the result. 
Just type:

> 1 + 2:

3



First Steps, 2

� If you want to store a value to a variable, type:

� Now the value 25 is stored to the name c, and you can refer to this 
number through the name c in subsequent calculations.

� Suppose that c is 25°Celsius. If you want to convert it to Fahrenheit, 
enter:

� The cls statement clears the screen, restart clears all values, and bye 
quits the interpreter.

> c := 25;

> 1.8*c + 32:

77



Names & Assignment

� A name always begins with an upper-case or lower-case letter or an 
underscore, followed by one or more upper-case or lower-case letters, 
underscores or numbers in any order.

� Use the assignment operator := to store a value to a name.

� Delete a value by assigning it to null or use clear:

15

> a := 1;

> var1 := 'hello world';

> a := null;

> clear var1;



Assignment, 2

� Compound assignment is supported in three fashions:

16

> a := 1;

> inc a;

> a:

2

> a := 1;

> a +:= 1;

> a:

2

mod

div

mul

dec

inc

StatementFunction Compound

Addition +:=

Subtraction -:=

Multiplication *:=

Division /:=

Modulo %:=

> a := 1;

> inc a, 2;

> a:

3

> a := 1;

> a +:= 2;

> a:

3

> c := 1;

> a := c++;

> a, c:

1    2

> a := c--;

> a, c:

2    1

> c := 1;

> a := ++c;

> a, c:

2    2

> a := --c;

> a, c:

1    1



Data Types



Integral & Rational Numbers

� Numbers can be represented like in the following examples.

� Integers:

� More than one value can also be printed at one line:

� Rational numbers:

� Scientific notation:

> -1:

-1

> 0, 1, 1.0, 1, 1.0:

0       1       1       1       1

> 3.141592654, -1.0:

3.141592654     -1

> 10e-3, -1e3, 2.3e3:

0.01    -1000   2300



Complex Numbers

� There are two notations to represent complex numbers.

� The ! operator:

� The I operand:

� Real part:

� Imaginary part:

19

> 1!2, -1.1!-2, 3!0:

1+2*I   -1.1-2*I        3

> 1+2*I, -1.1-2*I, 3+0*I:

1+2*I   -1.1-2*I        3

> real(1+2*I):

1

> imag(1+2*I):

2



Arithmetic, 1

� Agena allows to mix rational and complex numbers in calculations.

� Addition, subtraction, multiplication, division, and integer division:

� Examples:

20

rational complex/mixed

2 + 3 2+3*I + 1!2

2 – 3 2 - 3+1*I

2 * 3 2!2 * 3-I

2 / 3 2!0 / 3!1

2 \ 3 2!0 \ 3!1

> 2+3, 2!0/3!1, 2 + 3!1:

5       0.6-0.2*I       5+I



Arithmetic, 2

� Modulus (for rational numbers only):

� Exponentiation with rational or integer power:

� Exponentiation with integer power only (faster):

21

> 2 % 3:

2

> 2 ^ 3.1, 2 ^ 3:

8.5741877002903 8

> 2 ** 3:

8



Strings, 1

� Strings can be enclosed in single or double quotes. There is no 
difference in meaning.

� Concatenation of two or more strings:

22

> 'this is a text':

this is a text

> "this is a text":

this is a text

> 'Hello ' & 'world':

Hello world

> a := ’Hello ’;

> a &:= ’World’;

a:

Hello World



Strings, 2

� Substrings:

23

> str := 'abcd';

> str[2]:

b

> str[2 to 3]:

bc

> str[2 to -1]:   # from 2nd to last character

bcd

> str[-1]:        # last character

d

> str[-2 to -1]:  # last two characters

cd



Boolean Expressions & Relations, 1

� Agena supports the logical values true and false, also called `Booleans`.  
A third Boolean constant named fail indicates an error.

� Any condition, e.g. a < b, results to one of these logical values.

� Relational operators are:

24

Relation Operator

less than <

greater than >

less or equal <=

greater or equal >=

equality =

inequality <>



Boolean Expressions & Relations, 2

� Logical operators are:

25

Relation Operator

Boolean and and

Boolean or or

Boolean complement not

> 1 < 2:

true

> 1 < 2 and 1 = 0:

false

> true xor false:

true

norBoolean nor

Relation Operator

Boolean nand nand

Boolean exclusive-or xor



Boolean Expressions & Relations, 3

� If you add, subtract, multiply or divide a Boolean - or a relation that 
evaluates to a Boolean - with a number, then `true` will represent 
number 1 and `false` 0. 

� Thus, you can comfortably write statements without having to use `if` 
conditions, for example: 

26

> return (x > 0)*x;

> return if x > 0 then x else 0 fi;



Tables, 1

� Tables are used to represent more complex data structures. Tables 
consist of zero, one or more key-value pairs: the key referencing to the 
position of the value in the table, and the value the data itself.

� Tables can contain other tables, as well.

� To get the data with key 1, input:

27

> tbl := [

>    1 ~ ['a', 7.71],

>    2 ~ ['b', 7.70],

>    3 ~ ['c', 7.59]

> ];

> tbl[1]:

[a, 7.71]

> tbl := [

>    ['a', 7.71],

>    ['b', 7.70],

>    ['c', 7.59]

> ];

short formlong form



Tables, 2

� To get the second entry in the subtable, enter:

� There are two forms to create empty tables.

� Tables can even be nested:

� The size operator returns the size of a table or any other structure.

28

> tbl[1, 2]:

7.71

> tbl := [];

> create table tbl;

> [1, [2, [3]]]:

[1, [2, [3]]]



Tables, 3

� To select a sequence of elements in a table, use the to notation:

� When trying to index a null value with square brackets, Agena returns an 
error. When using curly brackets, however, Agena just returns null.

29

> tbl[1 to 2]:

[[a, 7.71], [b, 7.7]]

> tbl := null;

> tbl[1]:

Error in stdin at line 1:

attempt to index global `tbl` (a null value) with a number value

Stack traceback:

stdin, at line 1 in main chunk

> tbl{1}, tbl{1 to 2}:

null   null



Arrays

� Tables with positive integral keys are called arrays.

� Values can be inserted into arrays in two ways:

� Values can be deleted like this:

> tbl := [10, 11, 12];

> tbl[4] := 'a'; tbl[5] := 'b';

> insert 'a', 'b' into tbl;

> tbl[1] := null;

> delete 'a', 'b' from tbl;



Dictionaries

� Another form of a table is the dictionary which indices can be any kind of 
data - not only positive integers. Key-value pairs are entered with quoted 
keys and tildes, or with unquoted names and =. 

� As with arrays, indexed names are used to access the corresponding 
values stored to dictionaries.

� If a table key is a string, you can also use the notation:

31

> dic := ['donald' ~ 'duck', mickey = 'mouse'];

> dic['donald']:

duck

> dic.donald:

duck



Sets, 1

� Sets are collections of unique items: numbers, strings, and any other 
data except null. Any item is stored only once.

� If you want to check whether 'donald' is part of the set s, just index it as 
follows:

> s := {'donald', 'mickey', 'donald'}:

{donald, mickey}

> s['donald']:

true

> s['daisy']:

false



Sets, 2

� If you want to add or delete items to or from a set, use the insert and 
delete statements.

� The in operator also checks whether an item is part of a set.

� Sets consume around 40 % less memory than tables.

> insert 'daisy' into s;

> delete 'daisy' from s;

> 'donald' in s:

true

> 'daisy' in s:

false



Sequences, 1

� Sequences can hold any number of items except null.

� You can access the items the usual way:

� Values can be added as with tables.

34

> s := seq(1, 1, 'donald', true):

seq(1, 1, donald, true)

> s[2]:

donald

> s[4] := {1, 2, 2};

> insert [1, 2, 2] into s;



Sequences, 2

� Items can be deleted by setting their index position to null, or by applying 
delete.

� The in operator checks whether a sequence contains a given item.

� Sequences are twice as fast when adding values than tables.

> s[4] := null;

> delete [1, 2, 2] from s;

> 'donald' in s:

donald



Registers, 1

� Registers are fixed-size arrays that also can store nulls.

� You can access the items the usual way:

� If a value is deleted, the size of the register will not change:

36

> r := reg(null, 1, 'donald', true):

reg(null, 1, donald, true)

> r[3]:

donald

> r[2] := null;

> r:

reg(null, null, donald, true)



Registers, 2

� Registers have a pointer to the top of a register that can be changed so 
that data above the value of the top pointer can be hidden:

� Registers can be created with a predefined number of elements:

� The size of a register can be changed with the registers.reduce and 
registers.extend functions.

37

> registers.settop(r, 3); print(r, registers.gettop(r));

reg(null, null, donald) 3

> create register r(8);

> r:

reg(null, null, null, null, null, null, null, null)



Pairs

� Pairs hold exactly two values of any type (including null and other pairs).

� The left and right operators provide read access to its left and right 
operands; the standard indexing method using integers is supported, as 
well:

� The left and right operand of a pair can be changed as follows:

38

> p := 10:11;

> left(p), right(p), p[1], p[2]:

10      11      10      11

> p[1] := -10;



Write-Protection

� The freeze function write-protects a table, set, sequence, register, pair or 
userdata from modification.

� The unfreeze function removes the write-protection again.

Control Statements



if Statement, 1

� Conditions can be checked with the if statement. The elif and else 
clauses are optional. The closing fi is obligatory.

� A short form is also available if only one statement shall be executed 
and no else clause is needed:

41

> if 1 < 2 then

>    print('valid')

> elif 1 = 2 then

>    print('invalid')

> else

>    print('invalid, too')

> fi;

valid

> 1 < 2 ? print('valid')

valid



if Statement, 2

� If statements also support simple assignments in the conditions, even in 
elif clauses. If the right-hand side evaluates to neither false, fail nor null, 
then the corresponding then part will be executed.

� Compare:

42

> flag := io.read();

> if flag then

>    print(flag)

> fi;

> if flag := io.read() then

>    print(flag)

> fi;



if Statement, 3

� Assignments and the actual check can be combined in the if clause:

43

> if lnx := ln(1), lnx >= 0 then

>    print(lnx)

> fi;

0



if Operator, 1

� The if operator checks a condition and returns the result:

� An optional preceding with clause allows to define one or more auxiliary 
variables that are local to this operator only:

� You can also add one or more elif clauses.

44

> result := if 1 < 2 then 'valid' else 'invalid' fi;

> result:

valid

> x := Pi;

> a := with n := 2*x -> if x < 0 then n else 2*n fi;

> b := with m, n := x, 2*x -> if x < 0 then m else n fi;



if Operator, 2

� The extended version of the if operator is similar to the if statement. Note 
the sequence `if is` and the obligatory return expressions in the bodies; 
elif’s and else’s are optional, as are the statements in the bodies.

45

> a := 10;

> sgn := if is a < 0 then  # determines sign of `a'

>           print('I am negative');

>           [further statements ...]

>           return -1

>        elif a = 0 then

>           print('I am zero');  # just one statement

>           return 0

>        else                    # no statement 

>           return 1

>        fi;

> sgn:

1



case Statements, 1

� The case statement facilitates comparing values and executing 
corresponding statements.

46

> c := 10;

> case c

>    of -1 then        # one value to be compared

>       print('negative')

>    of 0, 1 then      # multiple values to be compared

>       print('non-negative')

>    of 2 to infinity  # a range

>       print('non-negative, too')

>    else

>       print('negative, too')

>    esle              # this keyword is optional, just a beautifier 

> esac;

non-negative, too



case Statements, 2

� A variant works like the if statement and may improve readability of 
code.

47

> x := 10;

> case  

>    of x < 0 then return -1

>    of x = 0 then return 0  

>    else return 1  

> esac 

1



onsuccess Clause

� Both if and case statements support an optional onsuccess clause. If at 
least one of the conditions evaluated to true, then the statements in the 
onsuccess clause are also executed.

48

> c := 'agena'; flag := false;

> case c

>    of 'agena' then

>       print('Agena !')

>    of 'lua' then

>       print('Lua !')

>    onsuccess

>       flag := true

>    else

>       print('Another programming language !')

> esac;

Agena !

> flag:

true



Alternative end Clause

� You can now use the end token instead of the closing fi, od, esac, yrt

and epocs keywords. Examples:

49

> if 1=1 then print(true) else print(false) end;

> for i to 10 do

>    print('Agena !')

> end;



Loops



for Loops, 1

� A for loop iterates over one or more statements.

� A numeric for loop begins with an initial numeric value (from clause), and 
proceeds up to and including a given numeric value (to clause). The step 
size can also be given (step clause). The od keyword indicates the end 
of the loop body.

� The current iteration value is stored to a control variable (i in this 
example) which can be used in the loop body.

51

> for i from 1 to 3 by 1 do

>    print(i, i^2, i^3)

> od;

1 1 1

2 4 8

3 9 27



for Loops, 2

� The from and step clauses are optional.

� If the from clause is omitted, the loop will start with the initial value 1.

� If the step clause is omitted, the step size will be 1.

52

> for i to 3 do

>    print(i, i^2, i^3)

> od;

1 1 1

2 4 8

3 9 27



for Loops, 3

� The value of the control variable can be accessed outside the loop.

� Since after the last iteration, the control variable is internally increased 
by the step size a very last time, its contents is:

53

> for i to 3 do

>    result := i^2

> od;

> i:

4



for Loops, 4

� A for/in loop iterates over all values in a table, set, and sequence. With 
strings, it iterates over each character from the left to the right.

54

> for i in ['Agena', 'programming', 'language'] do

>    print(i)

> od

Agena

programming

language

> for i in 'Agena' do print(i) od

A

g

e

n

a



for Loops, 5

� You can also iterate over the keys of a table (or sequence) or both keys 
and values:

55

> for keys i in ['donald' ~ 'duck', 'daisy' ~ 'duck'] do

>    print(i)

> od;

daisy

donald

> for i, j in ['donald' ~ 'duck', 'daisy' ~ 'duck'] do

>    print(i, j)

> od;

daisy   duck

donald  duck



while Loops, 1

� A while loop first checks a condition and if this condition is true or any 
other value except false, fail, or null, it will iterate the loop body again 
and again as long as the condition remains true.

� The following statements calculate the largest Fibonacci number less 
than 1000.

56

> a := 0; b := 1;

> while b < 1000 do

>    c := b; b := a + b; a := c

> od;

> c:

987



while Loops, 2

� A simple assignment can also be done in the while condition. This allows 
for shorter code. If the right-hand side evaluates to neither false, fail or 
null, then the loop body will be executed.

� Just compare the following two statements.

57

> flag := true;

> while flag do

>    flag := io.read();

>    if flag = 'Z' then break fi

> od

> while flag := io.read() do

>    if flag = 'Z' then break fi

> od



while Loops, 3

� You can combine an assignment and a conditional check in the while 
clause: When doing so, the assignment is redone each time flow control 
returns to the top of the loop, and the condition is checked again, as 
well.

58

> i := 0.3;

> while logn := ln(i), logn < -0.9 do

> print(i, logn); i +:= 0.1

> od;

0.3     -1.2039728043259

0.4     -0.91629073187416



do .. as & do .. until  Loops

� Variations of while are the do .. as and do .. until loops which check a 
condition at the end of the iteration.

� Thus – contrary to while loops - the loop body will always be executed at 
least once.

59

> c := 0;                          c := 0

> do                               > do

>    inc c                         >    inc c

> as c < 10;                       > until i = 10;

> c:                               > c:

10                                 10



do .. od Loops

� Infinite loops are Support by do .. od loops, a syntactic sugar for `while 
true do .. od`.

� See the `Loop Control` sheet on how to exit these loops.

60

> c := 0;

> do

>    inc c;

>    if c > 9 then break fi

> od;

> c:

10



Combined for/while Loops

� All flavours of for loops can be combined with a while condition. As long 
as the while condition is satisfied, i.e. is true, the for loop iterates.

� Likewise, the until condition quits the loop:

61

> for x to 10 while ln(x) <= 1 do 

>    print(x, ln(x)) 

> od;

1 0

2 0.69314718055995

> for x to 10 until ln(x) > 1 do 

>    print(x, ln(x)) 

> od;

1 0

2 0.69314718055995



for/until and for/as Loops

� for loops can also be combined with a closing until or as condition.

62

> for x to 10 do 

>    print(x) 

> as x < 3;

1

2

3

> for x to 10 do 

>    print(x) 

> until x = 3;

1

2

3



Conditional for Loops

� This variant initialises a new local control variable, checks a while or until 
condition and then executes the loop body. 

� You have to explicitly change the loop control variable in the body which 
allows for adaptive step sizes during a computation, for example when 
examining oscillatory functions with varying interval lengths.

63

> for i := 1 while i <= 3 do print(i); i +:= 1 od

1

2

3

> for i := 1 until i = 4 do print(i); i +:= 1 od

1

2

3



Loop Control, 1

� Agena features three statements to control loop execution. The following 
two are applicable to all loop types.

� The skip statement causes another iteration of the loop to begin at 
once, thus skipping all of the following loop statements after the skip 
keyword for the current iteration.

� The break statement quits the execution of the loop entirely and
proceeds with the next statement right after the end of the loop.

64

> for i to 5 do

>    if i = 3 then skip fi;

>    print(i);

>    if i = 4 then break fi

> od;

1

2

4



Loop Control, 2

� skip and break can also be combined with when or unless conditions:

65

> for i to 5 do

>    skip when i = 3;

>    print(i);

>    break when i = 4

> od;

1

2

4

> for i to 5 do

>    skip unless i <> 3;

>    print(i);

>    break unless i <> 4

> od;

1

2

4



Loop Control, 3

� The redo statement restarts the current iteration of a for/to or for/in loop 
from its beginning, without incrementing the loop control variable or 
processing the next item in a structure.

66

> flag := true;

> for i to 3 do

>    print(i);

>    if flag and i = 2 then 

>       flag := false; 

>       redo 

>    fi

> od;

1

2

2

3



Loop Control, 4

� The relaunch statement, however, restarts a for/to or for/in loop 
completely.

67

> flag := true;

> for i to 3 do

>    print(i);

>    if flag and i = 2 then 

>       flag := false; 

>       relaunch 

>    fi

> od;

1

2

1

2

3



Procedures



Procedures, 1

� Let us write a procedure to compute the factorial of an integer.

� A procedure can call itself to generate the final result.

� The return statement passes the result of a computation.

� The procname keyword is substituted by the name with which the 
procedure was invoked.

69

> factorial := proc(n) is  # factorial of an integer

>    if n < 0 then return fail

>    elif n = 0 then return 1

>    else return procname(n-1)*n

>    fi

> end;

> factorial(4):

24



Procedures, 2

� Alternatively, a function can be defined with the procedure statement.

You can put the local keyword before the procedure keyword to define 
local procedures.

70

> procedure factorial (n) is  # factorial of an integer

>    if n < 0 then return fail

>    elif n = 0 then return 1

>    else return procname(n-1)*n

>    fi

> end;

> factorial(4):

24



Local Variables

� A local variable is known only to the respective procedure and the block 
where it has been declared. 

� It cannot be used in other procedures, the interactive Agena level, or 
outside the block where it has been declared.

71

> factorial := proc(n) is

>    local result;

>    result := 1;

>    for i from 1 to n do result := result * i od;

>    return result

> end;

> factorial(10):

3628800



Variable Number of Arguments

� If you want to pass a variable number of arguments, use the ? keyword 
in the parameter list.

� The varargs system table contains all variable arguments passed with 
the ? facility. Values can be accessed like with any other table. 

� The system variable nargs contains the number of arguments passed 
(both with the ? facility and without).

72

> f := proc(?) is

>    return nargs, varargs, varargs[1]

> end;

> f('Beowulf', 'Grendel'):

2       [Beowulf, Grendel]      Beowulf



Options, 1

� A function does not have to be called with exactly the number of
parameters given at procedure definition. 

� You may optionally pass less or more values at run-time. If no value is 
passed for a parameter, then this parameter will automatically be set to 
null at function call.

� If you pass more arguments than there are actual parameters, excess 
arguments will be ignored.

> f := proc(a, b, c) is

>    return a, b, c

> end;

> f(1):

1       null    null



Options, 2

� Let us build an extended square root function that either computes in the 
real or complex domain. By default, i.e. if only one argument is given, the 
real domain will be chosen, otherwise you may explicitly set the domain 
using a pair as a second argument.

74

> xsqrt := proc(x, mode) is

>    if nargs = 1 or mode = 'domain':'real' then

>       return sqrt(x)

>    elif mode = 'domain':'complex' then

>       return sqrt(x + 0*I)

>    else

>       return fail

>    fi

> end;

> xsqrt(-2):

undefined

> xsqrt(-2, 'domain':'real'):

undefined



Options, 3

� If the left-hand value of the pair in a function call shall denote a string,
you can spare the single quotes put between the string by using the =

token which converts the left-hand name to a string.

> xsqrt(-2, domain = 'complex'):

1.4142135623731*I



Error Handling & Error Traps

� The error function issues an error:

� The try/catch statement catches errors:

� Alternatively, the protect function traps errors, as well.

> success, s := true, null;

> try

>    print(s[1])  # provoke an error by indexing null

> catch in msg then

>    success := false

> yrt;

> success:

false

> if 1 = 1 then error('Oops !') fi

Oops !

Stack traceback: in `error`

stdin, at line 1 in `(null)` in `(null)`



Type Checking, 1

� You can check the type of arguments passed in two ways:

� Query the type with the :: or :- (the negation) operators:

� State the expected type in the parameter list:

77

> f := proc(x) is

>    if x :- number then error('no number argument') fi; 

>    return x

> end;

> f('men ne cunnon hwyder helrunan hwyrftum scriþað.'):

wrong type of argument

> procedure f (x :: number) is

>    return x

> end;

> f('men ne cunnon hwyder helrunan hwyrftum scriþað.'):

Error in stdin:

invalid type for argument #1: expected number, got string.



Type Checking, 2

� Up to five types may be given when putting them in curly brackets:

� Besides checking the arguments, the return can also be validated:

78

> f := proc(x :: number) :: number is

>    return tostring(x)

> end

> f(1)

Error in stdin, at line 2:

`return` value must be of type number, got string.

> f := proc(x :: {number, complex}) is

>    return tostring(x)

> end

> f(1!2)

1 2



Type Checking, 3

� Numbers can be examined further with the keywords

� integer (any integral number), 

� posint (positive integer), 

� nonnegint (nonnegative integer), 

� positive (positive floats and integers), 

� nonnegative (nonnegative floats and integers).

> f := proc(x :: integer) is

>    return x

> end

> f(Pi)

Error in stdin:

type integer expected for argument #1, got number.



Type Checking, 4

� Function arguments can be checked further with the pre clause …

� … and the result with the post clause:

80

> f := proc(x :: number) is

>    return post x > 0 with x

> end

> f(0)

In stdin at line 2:

Error in post-condition: invalid return.

> f := proc(x :: number) is

>    pre x > 0 is

>    return x

> end

> f(0):

In stdin at line 2:

Error in pre-condition: posture not satisfied.



Predefined Results

� Predefined results can be set with the rtable.defaults function by entering 
them into a remember table.

� Agena will return the given predefined result if it exists and does not 
compute it by executing the procedure body, so there is also an increase 
in speed.

> rtable.defaults(fact, [  # defaults for fact(0) .. fact(3)

>    -1~undefined, 0~1, 1~1, 2~2, 3~6

> ]);

> fact(-1):

undefined

> rtable.defaults(fact):

[[2] ~ [2], [1] ~ [1], [0] ~ [1], [3] ~ [6], [-1] ~ [undefined]]



Efficient Recursion

� Agena will remember procedure results if the rtable.remember function is 
invoked. An optional table of predefined results can also be given. This 
speeds up recursive procedures significantly.

� For the differences between defaults and remember, check the manual 
(Chapter 14.4). Chapter 6.18.1 describes the feature reminisce shortcut.

> fib := proc(n) is

>    assume(n >= 0);

>    return fib(n-2) + fib(n-1)

> end;

> rtable.remember(fib, [0~1, 1~1]);

> fib(50):

20365011074



State Tables

� A table can be assigned to a function with the store feature. This internal 
table is available during a whole session and you can read from and 
write values to it in subsequent calls to the function.

> add := proc(x) is

>    feature store;

>    if x = null then  # set default value zero

>       store[1] := 0

>    else

>       store[1] +:= x

>    fi;

>    return store[1]

> end;

> add():

0

> add(10):

10



Functions as Binary Operators

� An ordinary function of two arguments can be called just like a binary 
operator.

� When using a function this way, it has always the highest precedence.

> plus := proc(x, y) is return x + y end;

> 1 plus 2:

3



Short-cut Procedures, 1

� If your procedure consists of exactly one expression, then you may use 
an abridged syntax if the procedure does not include statements such as 
if, for, insert, etc.

� Let us define a simple factorial function with one argument.

� A function with two arguments:

85

> factorial := << (x) -> exp(lngamma(x+1)) >>;

> factorial(4):

24

> sum := << (x, y) -> x + y >>;

> sum(1, 2):

3



Short-cut Procedures, 2

� The `with` clause allows to define local variables.

� Alternatively, you can define a function with the def or define statement:

86

> fact := << (x :: number)

>    with n := 1

>    -> exp(lngamma(x + n)) >>;

> fact := << (x :: number)

>    with m, n := 0, 1

>    -> exp(lngamma(x + n)) + m >>;

> define sum (x, y) -> x + y >>;

> sum(1, 2):

3



Object-Oriented Programming, 1

� Methods for tables can be implemented OOP-style using the @@ 
syntax:

87

> account := ['balance' ~ 0];

> proc account@@deposit(x) is

>    inc self.balance, x;

> end;

> account@@deposit(100)

> account.balance:

100

> proc account@@withdraw(x) is

>    dec self.balance, x

> end;



Object-Oriented Programming, 2

� A constructor that created new accounts:

88

> proc account@@new(o) is

>    setmetatable(o, self);

>    self.__index := self;

>    return o

> end;

> a := account@@new(['balance' ~ 0]);

> a.balance:

0



Object-Oriented Programming, 3

� Inheritance: here we define a new account class based on the one
defined above that does not allow overdrafts.

� For more information, please check Chapter 6.24 of the Primer and 
Reference.

89

> creditaccount := account@@new();

> proc creditaccount@@withdraw(x) is

>    if x > self.balance then error('Error, not enough credit.') fi;

>    dec self.balance, x;

>    return self.balance

> end;

> b := creditaccount@@new();

> b@@withdraw(1000):

Error, not enough credit.



Functional-Style Programming, 1

� There are some functional-programming-style functions and operators 
that spare you some lines of code, see Chapter 6.32 in the Primer & 
Reference:

90

combines two structures of equal sizezip

approximates seriesaddup

computes productsmulup

generates structuresforeach

applies a function on intermediate resultstimes

deletes all the elementsremove

applies a function on every item of a structuremap, @

Function Features

select, $ returns all the elements that satisfy a Boolean condition



Functional-Style Programming, 2

91

apply a function on each item of a structure or string and return 
an accumulated - that is summed-up - result

reduce, fold

maps one or more functions on a structurepipeline

transforms a function with multiple arguments into a sequence of
single-argument functions

factory.
curry

Function Features



with and Related Statements, 1

• The with statement unpack table values, indexed by string keys, declare 
them local and then access them in the respective block. After leaving 
the block, all the values listed right between the with and in tokens are 
automatically written back to the table.

> zips := [duedo = 40210, bonn = 53111, cologne = 50667];

> with duedo, cologne in zips do

>    duedo   := 40237;  # change duedo entry

>    cologne := null    # cologne will be deleted ☺

> od;

> zips:

[duedo = 40237, bonn = 53111]

> duedo, bonn in zips; # equals duedo, bonn := zips.duedo, zips.bonn

> duedo, bonn:

40237   53111



with and Related Statements, 2

� A flavour of the with statement allows to reference an entry by just an 
underscore. It also allows to actively change values in the table.

93

> zips := [duedo = 4000, bonn = 5300]

> with zips do  

>    print(_.bonn);  

>    _.bonn := 53111  

> od  

5300  

> zips:  

[bonn ~ 53111, duedo ~ 4000] 



with and Related Statements, 3

� Yet another variant allows to easily define local variables to be used in a 
block:

94

> with a, b := 1, 2 do

>    c := a + b

> od;  

> print(a, b, c):  

null   null   3



Syntactic Sugar

� Just an overview of some syntactic sugar available:

95

> break when x <> 0;

> if x <> 0 then break fi;

> skip when x <> 0;

> if x <> 0 then skip fi;

> return when x <> 0;

> if x <> 0 then return fi;  

> return when x <> 0 with y;

> if x <> 0 then return y fi;

> break unless x = 0;

> if x <> 0 then break fi;

> skip unless x = 0;

> if x <> 0 then skip fi;

> return unless x = 0;

> if x <> 0 then return fi;  

> return unless x = 0 with y;

> if x <> 0 then return y fi;



Printing, 1

� The print function writes values - numbers, strings, Booleans, tables, etc. 
to the screen:

� The printf function gives more control on the output format. In the 
following example %d depicts an integer and %f a float.

96

> print('sqrt(', 2, ') = ', sqrt(2)):

sqrt(   2       ) =     1.4142135623731

> print('sqrt(' & 2 & ') = ' & sqrt(2)):

sqrt(2) = 1.4142135623731

> printf('sqrt(%d) = %f', 2, sqrt(2)):

sqrt(2) = 1.414214



Printing, 2

� To print 10 decimal (fractional) places of sqrt(2), we put .10 in front of the 
f specifier:

� The %s formatter represents a string and %18.15f depicts a number with 
a total of 18 digits (pre-decimal places plus the decimal dot plus the 
fractional places), including 15 fractional places:

� For more information and examples, check Chapter 3.19 ‘Print Values’ in 
the Primer and Reference.

> printf('sqrt(%d) = %.10f', 2, sqrt(2)):

sqrt(2) = 1.4142135624

> printf('%s(%d) = %18.15f', 'sqrt', 2, sqrt(2)):

sqrt(2) =  1.414213562373095



Did you know ?



Did you know, 1 ?

� If you do not like the default prompt, just enter something like: 
_PROMPT := '% '

� You can load your own programmes into an Agena session by using the 
run function (e.g. run 'progname.agn') or starting Agena from the 
shell with agena –i progname.agn.

� If you want your self-written procedures, constants, etc. to be available 
every time you invoke the interpreter, just put them into an agena.ini file 
residing in your home directory.

� Data you compute in a session can be stored to a file using the save 
function, to be read into subsequent session later by read.

� You can send and receive data on the TCP level across the Internet and 
LANs with the net package.

Did you know, 2 ?

� Data stored in CSV and XML files can be imported with the utils.readcsv 
and utils.readxml functions. See xml package, too.

� The way Agena outputs tables, sets, sequences, complex numbers, and 
pairs can be changed by modifying the environ.aux.print* procedures in 
the library.agn file located in the lib directory of your Agena installation.

� On some 64-bit flavours of Windows 2003 Server and Windows 2008 
Server you may need to set the agena.exe binary file to Windows 2000 
or Windows XP compatibility mode in order for the interpreter to start 
successfully.

100



Miscellaneous



Precedence

� Operator precedence follows the table below, from lowest to highest.

Prec Operators

10 or xor nor xnor

9 and nand

8 < > <= >= = == <> ~= ~<> :: :- |

7 in notin subset xsubset union minus intersect atendof |-

6 & : @ $ $$

5 + - || ^^ split inc dec

4 * / % \ <<< >>> <<<< >>>> && *% /% +% -% %% symmod mul

div intdiv mod

3 not - +++ ---

2 ^ **

1 ! ~~ and all other unary operators



Mathematical Operators

103

Division/

Exponentiation with rational power^

Addition+

Integer division\

Operator Description

- Subtraction

* Multiplication

** Exponentiation with integral power

% Modulus



Mathematical Functions, 1

104

Hyperbolic sinesinh(x)

Inverse cosinearccos(x)

Inverse tangentarctan(x)

Secantsec(x)

Cosecantcsc(x)

Sinesin(x)

Hyperbolic cosinecosh(x)

Function Description

cos(x) Cosine

tan(x) Tangent

cot(x) Cotangent

arcsin(x) Inverse sine



Mathematical Functions, 2

105

Logarithm of x to base blog(x, b)

Exponentiation e^xexp(x)

Natural logarithmln(x)

Inverse hyperbolic tangentarctanh(x)

Cardinal sinesinc(x)

Hyperbolic tangenttanh(x)

Square rootsqrt(x)

Function Description

arcsinh(x) Inverse hyperbolic sine

arccosh(x) Inverse hyperbolic cosine

cosc(x) Cardinal cosine

tanc(x) Cardinal tangent



Mathematical Functions, 3

106

Signsign(x)

Error functionerf(x)

Absolute value/magnitudeabs(x)

Hypotenusehypot(x, y)

Gamma functiongamma(x)

Cubic rootcbrt(x)

Rounds x downwards to the nearest integerentier(x)

Function Description

root(x, n) Non-principal n-th root of x

proot(x, n) Principal n-th root of x

lngamma(x) Logarithmic Gamma function

fact(n) Factorial



Mathematical Functions, 4

107

Rounds downwards to the nearest integer (same as entier)floor(x)

Rounds x to d-th digitround(x, d)

Checks for even numbereven(x)

Rounds upwards to the nearest integerceil(x)

Function Description

int(x) Rounds to the nearest integer towards zero

frac(x) Fractional part

odd(x) Checks for odd number



Mathematical Constants

� Agena features the following numeric constants, some of them are:

108

Equals 1.084202172485504434e-19DoubleEps

Golden ratio (1 + sqrt(5))/2Phi

Equals 3.14159265358979323846Pi

180/Pi to convert radians to degreesdegrees

Constant Meaning

Eps Equals 1.4901161193847656e-08

radians Factor Pi/180 to convert degrees to radians

Exp Constant e = exp(1) = 2.71828182845904523536

I Imaginary unit

infinity Infinity

undefined An expression stating that it is undefined, e.g. a singularity

See also Chapter A3of the Primer and Reference.

5

5



String Functions & Operators, 1

109

Matches patterns including ? and * wildcardsstrings.glob

Looks for the match of a patternstrings.match

Searches for a substringin

Replaces substringsstrings.replace

Inserts a substringstrings.include

Number of occurrences of a substring patternstrings.hits

Searches for a substring, supports pattern matchingstrings.find

Concatenation operator&

Checks whether a substring is not includednotin

Function Description

atendof Checks whether a string ends in the given pattern 

strings.remove Removes a substring



String Functions & Operators, 2

110

Converts a number to a stringtostring

Extracts given fields (columns) in a stringstrings.fields

Remove all leading/trailing white spaces or a given character 
or string

strings.ltrim
strings.rtrim

Removes leading and trailing white spacesstrings.trim

Converts a string to a numbertonumber

C-style formattingstrings.format

ASCII code conversionabs, char

Splits a string into its wordssplit

String lengthsize

Checks for an empty/filled stringempty/filled

Function Description



String Functions & Operators, 3

111

Searches a string with a regular expressionregex.find

Counts the number of matchesregex.count

Defines a regular expression patternregex.new

Shannon entropy indicatorsstrings.shannon

Converts to lower/upper case
strings.lower/
strings.upper

Damerau-Levenshtein distance of two stringsstrings.dleven

Capitalises/uncapitalises a string
strings.capitalise/

strings.uncapitalise

Compares two strings case-insensitively and returns an 
estimate of their similarity

strings.fuzzy

Function Description

strings.jaro Jaro(-Winkler) similarity of two strings



String Functions & Operators, 4

112

Looks for the first match of a regex-pattern in a stringregex.match

Function Description



Packages, 1

� Agena features various packages.

113

Bloom filter for strings and numbersbloom

Functions to process hours, minutes, and secondsclock

Bi-directional mapsbimaps

GNU iconv port, to transform strings between codepagesaconv

Bits and bytes twiddlingbytes

Package Function

ads Database specialised on storing and retrieving strings

bags Multisets, Cantor sets that count occurrences

astro Astronomical time and date functions

binio Functions for processing binary files

calc Undergraduate Calculus package



Packages, 2

114

Cuckoo filter for stringscuckoo

(n)curses binding to build terminal applicationscurses

RS-232 communication via COM portscom

Combinatorial functionscombinat

Physical unit conversion (lengths, weights, etc.)convert

CORDIC numeric functionscordic

Fractionsdiv

Dual numbersdual

Access to the Agena environmentenviron

Numeric approximationsfastmath

Package Function

fractals Various fractals & plotting routines, some FRACTINT support



Packages, 3

115

JSON encoding & decodingjson

INI file encoding & decoding (iniparser binding)ini

Fast Fourier Transformkiss

Fuzzy string matchingfzy

Graphicsgdi

Skewed & binary heaps plus AVL treesheaps

Read and write UNIX gzip compressed filesgzip

String and number hasheshashes

Input/output functions for console and filesio

Undergraduate Linear Algebralinalg

Linked listsllist

Package Function



Packages, 4

116

Aliases to Maple functionsmaple

Numeric C arraysnumarray

80-Bit Floating-Point arithmeticlong

Lookup tableslookup

Mathematical arbitrary precision librarymapm

Additional mathematical functionsmath

String memory filesmemfile

GNU Multiple Precision Arithmetic Library (GMP)mp

GNU Multiple Precision Floating-Point Reliable Library (MPFR)mpf

IPv4-based exchange of data over the Internet or LANs net

Number Theorynumtheory

Package Function



Packages, 5

117

Functions to operate with the underlying operating systemos

Red-black binary treesrbtree

Regular expression matching (PCRE2)regex

Functions for register administrationregisters

Functions to access the registryregistry

Administration of remember tablesrtable

Unique integer IDssema

Utilities and easy-to-use wrappers to some functionsskycrane

Functions for stack operationsstack

Statistical functionsstats

Various string handling functionsstrings

Package Function



Packages, 6

118

Functions specialised on table processingtables

Functions to list, read, and extract UNIX tar archivestar

Utility functions, e.g. CSV import and exportutils

xBase file support (i.e. dBASE (tm) III+)xbase

XML encoding & decoding (LuaExpat port)xml

Sinclair ZX Spectrum numeric functionszx

Package Function



Excurse: Doing Math with Agena

� There are four packages for undergraduate mathematics:

� calc – calculus,

� linalg – linear algebra,

� stats – statistics,

� combinat – combinatorics,

� kiss – Fast Fourier Transform.

� With the exception of kiss, all these aforementioned packages are built-
in, so you do not need the import statement to load them into a session.

� As you will see, you actually do not have to know much about the
interpreter to do some undergraduate math. 

Calculus, 1

� Define a function, for example f(x) = sin(x):

� Print a table of values, with and without formatting:

� Determine all the zeros over [-5, 5]: 

120

> f := << x -> sin(x) >>

> for x from -1 to 1 by 0.5 do

> print(x, f(x))

> od;

-1 -0.8414709848079

-0.5 -0.4794255386042

0 0

0.5 0.4794255386042

1 0.8414709848079 

> for x from -0.5 to 0.5 by 0.25 do

> printf('%+05.2f %+10.6f\n', x, f(x))

> od;

-0.50 -0.479426

-0.25 -0.247404

+0.00 +0.000000

+0.25 +0.247404

+0.50 +0.479426 

> calc.zeros(f, -5, 5):

seq(-3.1415926535898, 0, 3.1415926535898) 



Calculus, 2

� Differentiate f at point 0:

� Evaluate the third derivative of f at point 0: 

� Compute the minimum and maximum values on the interval [-10, 10],

or try calc.extrema(f, -10, 10).

121

> calc.differ(f, 0):

1 

> calc.differ(f, 0, deriv=3):

-0.99999999999983 

> calc.minimum(f, -10, 10):

seq(-7.8539816339745, -1.5707963267949, 4.7123889803847)

> calc.maximum(f, -10, 10):

seq(-4.7123889803847, 1.5707963267949, 7.8539816339745)



Calculus, 3

� Determine points of inflection and saddle points:

� Integrate f over [0, Pi]:

� Compute the series `Sum(1/n!, n=0 .. 100)` to return an approximation of
Euler's number:

122

> calc.integ(f, 0, Pi): 

2 

> calc.fsum(<< n -> 1/fact(n) >>, 0, 100):

2.718281828459 

> calc.inflect(f, 0, Pi): 

seq(0, 3.1415926535897878, 6.2831853071795827, 9.4247779607693847)

> calc.saddles(<< x -> x^3 >>, -1, 1):

seq(0) 



Linear Algebra, 1

� Define two vectors in different fashions: In the simple form, just pass all 
components explicitly; or pass only the non-zero components: 

� Check whether a and b are parallel and have the same direction: 

� Set a vector component by indexing: 

123

> a := < 1, 2, 3 >:

< 1, 2, 3 >

> b := vector(3, [1 ~ 2]):

< 2, 0, 0 > 

> abs(a+b) = abs(a) + abs(b):

false

> b[3] := 1; 



Linear Algebra, 2

� Now read the modified vector and its rightmost component - a negative 
integral index n depicts the |n|-th element from the right: 

� Addition and subtraction: 

124

> b:

< 2, 0, 1 >

> b[3], b[-1]:

1 1 

> a + b:

< 3, 2, 4 >

> a - b:

< -1, 2, 2 > 



Linear Algebra, 3

� Scalar, dot and cross product: 

� The determinant:

125

> 2 * a:

< 2, 4, 6 >

> a * a:

14

> linalg.crossprod(a, b):

< 2, 5, -4 > 

> linalg.det(A):

-59 



Linear Algebra, 4

� Find the vector x which satisfies the matrix equation A x = b. The matrix
constructor expects row vectors. 

126

> A := < < 1, 2, -4 >, < 2, 1, 3 >, < -3, 1, 6 > >:

[ 1, 2, -4 ]

[ 2, 1, 3 ]

[ -3, 1, 6 ]

> b := < -6, 5, -2 >:

< -6, 5, -2 >

> linalg.linsolve(A, b):

< 2, -2, 1 > 



Statistics, 1

� First we define a distribution: 

� Minimum and maximum observations along with their position in the
distribution:

� Arithmetic mean:

127

> s := seq(10, 8, 1, 6, 5, 2, 9, 7, 3, 4):

seq(10, 8, 1, 6, 5, 2, 9, 7, 3, 4) 

> stats.min(s):

1 3

> stats.max(s):

10 1 

> stats.amean(s):

5.5 



Statistics, 2

� The median: 

� For the first quartile, the median and the third quartile of a distribution, 
along with the minimum, the maximum observation, and the arithmetic
mean, in this order, enter:

� Standard and absolute deviation: 

128

> stats.median(s):

5.5

> stats.sd(s):

2.872281323269

> stats.ad(s):

2.5 

> stats.fivenum(s):

seq(2.75, 5.5, 8.25, 1, 10, 5.5)



Statistics & Combinatorics

� Outliers: 

� The Cartesian product: 

� Combinations and number of combinations: 

129

> s := seq(-100, 8, 10, 1, 6, 5, 2, 9, 7, 3, 4, 1000):

seq(-100, 8, 10, 1, 6, 5, 2, 9, 7, 3, 4, 1000)

> stats.chauvenet(s):

seq(1000, -100) 

> combinat.cartprod([[1, 2, 3], [30], [50, 100]]):

[[1, 30, 50], [1, 30, 100], [2, 30, 50], [2, 30, 100], [3, 30, 50], 

[3, 30, 100]]

> combinat.chosse(3, 2):

[[1, 2], [1, 3], [2, 3]]

> combinat.numbcomb(3, 2):

3 



Combinatorics, 2

� Permutations and number of permutations:

> combinat.permute([1, 2, 3], 3):

[[1, 2, 3], [1, 3, 2], [2, 1, 3], [2, 3, 1], [3, 1, 2], [3, 2, 1]]

> combinat.numbperm(3, 2):

6 



Any Questions ?

� For further information, please consult 

� the Primer and Reference, 
a manual explaining Agena on 1240+ 
pages

� the Quick Reference, 
an overview of all the functions available

� Both are available at

http://sourceforge.net/projects/agena/Manuals/

(Take the last slash in its URL.)


